

1. Emulació senzilla d'una multiplicació per dos (només sumar un 1 a l'exponent, sense comprovar res més). Nombre en \$f12 i resultat en \$f0.

```
mults_2:mfc1 $t0, $f12      # copia $f12 en $t0
        li $t1, 0x00800000  # màscara: un un en la 23a posició
        add $t0, $t0, $t1    # suma un un a l'exponent
        mtc1 $t0, $f0       # copia el resultat en $f0
        jr $ra
```

Es pot demanar com a treball per a fer a casa l'ampliació de la subrutina perquè tinga també en compte els casos especials (zero, infinit i NaN). És una bona excusa perquè els alumnes treballen el format. Us l'afegim:

```
mults_2:mfc1 $t0, $f12      # copia $f12 en $t0
        li $t1, 0x7FFFFFFF  # màscara per a E i M
        and $t1, $t0, $t1    # extrau E i M en $t1
        beq $t1, $zero, eixir # si E=M=0 es +0.0 o bé -0.0

        li $t1, 0x7F800000  # màscara per a E
        and $t2, $t0, $t1    # extrau E en $t2
        beq $t1, $t2, eixir  # si exp==255 és NaN, +Infty o -Infty

        li $t1, 0x00800000  # màscara: un un en la 23a posició
        add $t0, $t0, $t1    # suma un un a l'exponent
        li $t1, 0x7F800000  # màscara per a E
        and $t2, $t0, $t1    # extrau E en $t2
        bne $t2, $t1, eixir  # si exp<>255 és un nombre "normal"
                                # si exp==255 ha desbordat i cal

generar un infinit
        li $t1, 0xFF800000  # màscara amb M a zero
        and $t0, $t0, $t1    # posa M del resultat a zero

eixir:  mtc1 $t0, $f0
        jr $ra
```

2. Subrutina que diu si un nombre real és un NaN. Nombre en \$f12 i resultat en \$v0.

```
es_NaN: mfc1 $t0, $f12      # copia $f12 en $t0
        li $t1, 0x7F800000  # màscara per a l'exponent
        and $t2, $t0, $t1    # exponent en $t2
        bne $t1, $t2, noes    # si exp<>255 no és NaN
        li $t1, 0x007FFFFF  # màscara per a la mantissa
        and $t2, $t0, $t1    # mantissa en $t2
        beq $t2, $zero, noes  # si mantissa==0 no és NaN

sies:   li $v0, 1
        j  fin
noes:   li $v0, 0
fin:    jr $ra
```

3. Subrutina que calcula la mitjana aritmètica dels elements d'un vector. En \$a0 l'adreça del vector i en \$a1 la dimensió; resultat en \$f0.

```
.data 0x10000000
A:    .float 39.8, 36.2, 41.7, 40.5

m_arit: or $t0, $zero, $a0      # copia de $a0 en $t0
        or $t1, $zero, $a1      # copia de $a1 en $t1
        mtc1 $zero, $f0         # posa 0.0 en $f0

bucle:  lwc1 $f4, 0($t0)         # llig element A[i]
        add.s $f0, $f0, $f4      # incrementa $f0 amb A[i]
        addiu $t0, $t0, 4        # actualitza adreça a A[i+1]
        addi $t1, $t1, -1        # decrementa comptador
        bne $t1, $zero, bucle    # salta si queden elements

        mtc1 $a1, $f4           # copia $a1 (dimensió) en $f4
        cvt.s.w $f4, $f4        # converteix la dimensió en real
        div.s $f0, $f0, $f4      # divideix entre la dimensió
        jr $ra
```