Subasta en Jason

Agentes Inteligentes (AIN)

Authors: Vicent Botti, Carlos Carrascosa, Vicente Julián

Tema 6- Indice

Ejemplo: Sistema de Subastas en JASON



Diagrama Resumen del Sistema



auction.mas2j

```
// crea un SMA llamado auction

MAS auction {

    infrastructure: Centralised

    agents: ag1; ag2; ag3;

            auctioneer agentArchClass AuctioneerGUI;

}
```

Diagrama Resumen del Sistema: Agente Subastador (*auctioneer.asl*)



auctioneer.asl (1/2)

```
//starts the auction and identify the winner

/* beliefs and rules */
all_bids_received(N)            :-
.count(place_bid(N,_),3).

/* plans */

// goal created by the GUI of the agent
+!start_auction(N) : true

 <- -+auction(N);
      -+winner(N, noone, 0);
      .broadcast(tell, auction(N)).
```

```
// receive bid and check for new winner
@pb1[atomic]
+place_bid(N,V)[source(S)]
   :  auction(N) & winner(N,CurWin,CurVl) & V > CurVl
   <- -winner(N,CurWin,CurVl);
      +winner(N,S,V);
      .print("New winner is ",S, " with value ",V);
      !check_end(N).

@pb2[atomic]
+place_bid(N,_) : true
   <- !check_end(N).
```

auctioneer.asl (2/2)

```
+!check_end(N)
  :  all_bids_received(N) & winner(N,W,Vl)
  <- .print("Winner is ",W," with ", Vl);
     show_winner(N,W); // show it in the GUI
     .broadcast(tell, winner(W));
     .abolish(place_bid(N,_)).

+!check_end(_).
```

Agent overview diagram: Agent ag1



ag1.asl

```
// this agent always bids 6

+auction(N)[source(S)] : true

  <- .send(S, tell, place_bid(N,6)).
```

Agent overview diagram: Agent ag2



ag2.asl

```
/* This agent usually bids 4,  when it has an alliance with ag3, it bids 0 */
default_bid_value(4).
ally(ag3).

+auction(N)[source(S)] : not alliance
  <- ?default_bid_value(B);
     .send(S, tell, place_bid(N,B)).

+auction(N)[source(S)] : alliance
  <- .send(S, tell, place_bid(N,0)).

// alliance proposal from another agent
+alliance[source(A)]
  :  .my_name(I) & ally(A)
  <- .print("Alliance proposed by ", A);
     ?default_bid_value(B);
     .send(A,tell,bid(I,B));
     .send(A,tell,alliance(A,I)).
```

Agent overview diagram: Agent ag3



ag3.asl (1/2)

```
/* this agent bids 3, if it looses 3 auctions, it proposes an alliance to ag2 and therefore it bids 7 (3
from itself + 4 from ag2) */
default_bid_value(3).
ally(ag2).
threshold(3).

+auction(N)[source(S)]  :  (threshold(T) & N < T) |
                           (.my_name(I) & winner(I) & ally(A) & not alliance(I,A))
  <- !bid_normally(S,N).

+auction(N)[source(S)] :  .my_name(I) & not winner(I) & ally(A) & not alliance(I,A)
  <- !alliance(A);
     !bid_normally(S,N).

@palliance
+auction(N)[source(S)]    :  alliance(_,A)
  <- ?default_bid_value(B);
     ?bid(A,C);
     .send(S, tell, place_bid(N,B+C)).
```

ag3.asl (2/2)

```
+!bid_normally(S,N) : true
        <- ?default_bid_value(B);
           .send(S, tell, place_bid(N,B)).

@prop_alliance[breakpoint]
+!alliance(A) : true
        <- .send(A,tell,alliance).
```

Protocol auction

Name	auction
Description	
Included messages	**new auction: auctioneer agent --> ag1, ag2, ag3,** **bid: ag1 --> auctioneer agent,** **bid: ag2 --> auctioneer agent,** **bid: ag3 --> auctioneer agent,** **winner announcement: auctioneer agent --> ag1, ag2, ag3**
Scenarios	
Agents	auctioneer agent, ag1, ag2, ag3

¿Qué tipo de subasta se está realizando? Sobre cerrado de primer precio

Modificar los agentes para realizar una subasta holandesa

Protocol alliance

Name	alliance
Description	
Included messages	**alliance proposal: ag3 → ag2,** **alliance accepted: ag2 → ag3**
Scenarios	
Agents	ag1, ag2

console

```
[auctioneer] New winner is ag1 with value 6
[auctioneer] Winner is ag1 with 6
[auctioneer] New winner is ag3 with value 3
[auctioneer] New winner is ag1 with value 6
[auctioneer] Winner is ag1 with 6
[auctioneer] New winner is ag2 with value 4
[ag2] Alliance proposed by ag3
[auctioneer] New winner is ag1 with value 6
[auctioneer] Winner is ag1 with 6
[auctioneer] New winner is ag1 with value 6
[auctioneer] New winner is ag3 with value 7
[auctioneer] Winner is ag3 with 7
```